VIBRATION ANALYSIS OF NON-HOMOGENEOUS BEAMS

by 1264

PARSHURAM GANESH DATE

B.E. (Mech)., Marathwada University
Aurangabad, India, 1964

A MASTER'S REPORT

submitted in partial fulfillment of the

requirements for the degree

MASTER OF SCIENCE

Department of Mechanical Engineering

KANSAS STATE UNIVERSITY
Manhattan, Kansas

1969

Approved by:

Hugh S. Walker

Major Professor

TABLE OF CONTENTS

LIST OF FIGURES

LIST OF TABLES

NOMENCLATURE

English symbols

a	slope of elastic modulus-temperature relationship divided by reference elastic modulus.
$a(x)$	coefficient of $(n-1)^{th}$ derivative in n^{th} order differential equation such as equation (14)
A_n	coefficients of linearly independent solutions in the general solution.
$b(x)$	coefficient of $(n-2)^{nd}$ derivative in equation (14).
C_{ij}	coefficients of powers of x in the power series solution.
E_o	elastic modulus at reference temperature.
$E(T)$	elastic modulus as a function of temperature.
$E(x)$	elastic modulus as a function of length.
$f(x)$	function representing variation of elastic modulus with length.
I	moment of inertia of the cross section of the beam.
L	length of beam.
r	root of the indicial equation.
t	time variable.
T	temperature.
T_o	reference temperature.
$T(x)$	temperature as a function of length.
$T(t)$	function of time as defined by equation (4).
U	deflection at any point x
$\bar{U}(\bar{x})$	function of axial coordinate representing deflection, defined by equation (4)
$\bar{y}(\bar{x},t)$	vertical displacement of beam.
x	non dimentional axial coordinate.

$\bar{x}$ actual length coordinate.

Greek symbols

α temperature parameter as defined by equation (12)

λ eigenvalues of the problem as defined by equation (7)

ρ mass density of the beam per unit length.

Ω^2 separation constant, square of natural frequency as defined by equation (7)

CHAPTER I

INTRODUCTION

The use of lighter structural materials in aerospace industry has created a need for analysing the vibration problem based on non-homogeneous elastic theory. The elastic modulus of these light structural materials varies considerably with temperature. Typically, the temperature gradient along the beam may be the result of aerodynamic heating.

Marangoni, Fauconneau and Scipio (1)* have analysed the effects of non-homogeneity on transverse vibrational frequencies of uniform beams. They have used upper and lower bounding techniques, the Rayleigh Ritz method for upper bounds and the method of second projection for lower bounds, to effectively bracket the eigenvalues. Walker and Huang (2) have analysed Vibration and Stability of rockets (tapered beam), which leads to the same type of mathematical model as the non-homogeneous beam. They have used Frobenius' method to get a series solution to the linear differential equation with nonconstant coefficients.

This report demonstrates the use of Frobenius' method for any vibration problem leading to a linear differential equation with nonconstant coefficients. The nonhomogeneous beam is taken as a typical case leading to such a differential equation formulation. In fact, most cases involving variation of

---

* Numbers in parentheses refer to the numbers in the list of references.

density, Moment of Inertia, modulus of elasticity etc. can be handled by this method. This method seems most direct and straightforward. Greater accuracy can be expected since, this method does not invole large matrix operations. Most other methods yield large matrices to work with, which cause round-off errors. The results obtained are comared with those obtained by Marangoni, Fauconneau and Scipio.

CHAPTER II

FORMULATION OF THE PROBLEM

Consider a uniform beam of length L and subjected to a steady temperature distribution $T(\bar{x})$ causing the modulus of elasticity to become a function $E(\bar{x})$ of $\bar{x}$. Experimental investigations on the variation of modulus of elasticity with temperature conducted by Garrick (4), Spinner (5) and Hoff (6) show that a linear relationship between elastic modulus and temperature provides a good correlation for wide temperature ranges for most engineering materials. The relationship can be given as

$$E(T) = E_o\left[1-a(T-T_o)\right] \tag{1}$$

Therefore, knowing $T(\bar{x})$, we can get $E(\bar{x})$ from equation (1). Neglecting the effect of rotary inertia, which is small for the lower modes and small amplitude of vibrations, the equation of motion for the beam can be written as

$$\frac{\partial^2}{\partial \bar{x}^2}\left[E(\bar{x})I\,\frac{\partial^2 \bar{y}}{\partial \bar{x}^2}\right] = -\rho\,\frac{\partial^2 \bar{y}}{\partial t^2} \tag{2}$$



Fig. 1

with the boundary conditions,

a) Simply Supported beam:

$$\bar{y} = 0, \quad \frac{\partial^2 \bar{y}}{\partial \bar{x}^2} = 0 \qquad \text{at } \bar{x} = 0 \text{ and } \bar{x} = L \tag{3a}$$

b) Beam with clamped ends:

$$\bar{y} = 0, \quad \frac{\partial \bar{y}}{\partial \bar{x}} = 0 \qquad \text{at } \bar{x} = 0 \text{ and } \bar{x} = L \tag{3b}$$

c) Beam with free ends:

$$\frac{\partial^2 \bar{y}}{\partial \bar{x}^2} = 0, \ \frac{\partial}{\partial \bar{x}} \left[E(\bar{x}) I \frac{\partial^2 \bar{y}}{\partial \bar{x}^2} \right] = 0 \text{ at } \bar{x} = 0 \text{ and } \bar{x} = L \tag{3c}$$

Depending on what $E(\bar{x})$ is, we will get a differential equation with variable coefficients. A similar formulation will result from variation in density and/or moment of inertia also.

Assuming a product solution of the form,

$$\bar{y}(\bar{x},t) = \bar{U}(\bar{x})\, T(t) \tag{4}$$

where $\bar{U}$ is a function of $\bar{x}$ alone and T is a function of t alone.

Substituting equation (4) in (2), we get a set of ordinary differential equations

$$\frac{d^2}{d\bar{x}^2} \left[\frac{E(\bar{x})\, I}{\rho} \frac{d^2 \bar{U}}{d\bar{x}^2} \right] - \Omega^2\, \bar{U}(\bar{x}) = 0 \tag{5}$$

and

$$\frac{d^2 T}{dt^2} + \Omega^2\, T(t) = 0 \tag{6}$$

from equation (6) we see that Ω, the constant parameter, is the frequency of vibration. Substituting $E_o f(\bar{x})$ for $E(\bar{x})$ in equation (5) we get,

$$\frac{d^2}{d\bar{x}^2}\left[\frac{E_o f(\bar{x})\ I}{\rho}\ \frac{d^2\bar{U}}{d\bar{x}^2}\right] - \Omega^2 \bar{U}(\bar{x}) = 0 \tag{5a}$$

It is convenient to transform equation (5a) into dimensionless form. Using the transformations,

$$x = \frac{\bar{x}}{L},\quad U = \frac{\bar{U}}{L},\quad \text{and}\quad \lambda = \frac{\Omega^2 L^4 \rho}{E_o I},$$

we get the formulation in dimensionless form,

$$\frac{d^2}{dx^2}\left[f(x)\ \frac{d^2U}{dx^2}\right] = \lambda U \tag{7}$$

with the boundary conditions,

a) Simply supported beam:

$$U = 0 \text{ and } U'' = 0^{*} \qquad \text{at } x = 0 \text{ and } x = 1 \tag{8}$$

b) Beam with clamped ends:

$$U = 0 \text{ and } U' = 0 \qquad \text{at } x = 0 \text{ and } x = 1 \tag{9}$$

c) Beam with free ends:

$$U'' = 0 \text{ and } \frac{d}{dx}\left[f(x)\ U''\right] = 0 \text{ at } x = 0 \text{ and } x = 1 \tag{10}$$

---

* Primes denote derivatives with respect to x

In the problem that is worked here, a linear temperature distribution along the length of the beam is considered. Nevertheless, other temperature distributions like parabolic, trignometric etc. can be handled along similar lines.

The temperature distribution along the length of the beam is as shown in Fig. 2.



Fig. 2

Mathematically, $T(x) = T_o + x(T_1 - T_o)$ (11)

Substituting equation (11) in (1) we get,

$$E(x) = E_o\left[1-a(T_1-T_o)x\right]$$

let $\alpha = a(T_1 - T_o)$, $0 \leqq \alpha < 1$ (12)

then $E(x) = E_o(1 - \alpha x) = E_o f(x)$

($\alpha = 0$ means that there is no temperature variation along the beam, and $\alpha = 1$ means that the temperature difference has reached a point of 'zero elasticity' which is a non-practicable case)

Substituting for f(x) in equation (7) we get,

$$\frac{d^2}{dx^2}\left[(1 - \alpha x)\,\frac{d^2U}{dx^2}\right] - \lambda U = 0$$

$$\text{i.e. } (1 - \alpha x)U'''' - 2\alpha U''' - \lambda U = 0 \tag{13}$$

with the appropriate boundary conditions such as equations (8), (9) and (10).

CHAPTER III

DISCUSSION OF THE SOLUTION

Frobenius' method (7) is used to obtain a series solution to the differential equation. Changing the formulation to the form,

$$U'''' + a(x)\,U''' + b(x)U = 0 \tag{14}$$

one can show easily that a(x) and b(x) are both analytic with radius of convergence $\frac{1}{\alpha}$. Therefore, the power series solution obtained is convergent for the entire range of x, $0 \leq x \leq 1$. (8)

It is interesting to note that a different formulation of the same problem with temperature distribution viewed as shown in Fig. 3 is,

$$[1 - \alpha(1 - x)]\,U'''' + 2\alpha U''' - U = 0 \tag{13a}$$

where $\alpha = a(T_o - T_1)$

$\lambda = \frac{\Omega^2 L^4 \rho}{E_1 I}$, E_1 is the elastic modulus at reference temperature T_1.



Fig. 3

The series solution by Frobenius' method is convergent only for $|x| < \frac{1-\alpha}{\alpha}$. Therefore, with this formulation, one cannot handle cases with $\alpha \geqq 0.5$. Sometimes changing the form of the equation helps in getting a desired type of solution (7).

Frobenius' method consists of assuming solutions of the form,

$$U_{r+1} = \sum_{m=1}^{\infty} C_{m,r+1} \; x^{m+r-1} \tag{15}$$

where r is a root of the indicial equation, and $C_{1,r+1} \neq 0$ (The summation is taken over the range 1 to ∞ in order to maintain close correspondance with subscripted variables in the computer program. The subscript r+1 also stems from the same reasoning.)

We will get as many roots as the order of the differential equation. Each root will give one linearly independent solution. Differentiating equation (15) with respect to x we get,

$$U' = \sum_{m=1}^{\infty} C_{m,r+1}(m+r-1) \; x^{m+r-2} \tag{16a}$$

$$U'' = \sum_{m=1}^{\infty} C_{m,r+1}(m+r-1)_2 \; x^{m+r-3} \tag{16b}$$

$$U''' = \sum_{m=1}^{\infty} C_{m,r+1}(m+r-1)_3 \; x^{m+r-4} \tag{16c}$$

and

$$U'''' = \sum_{m=1}^{\infty} C_{m,r+1}(m+r-1)_4 \; x^{m+r-5} \tag{16d}$$

where $(m+r-1)_n$ represents descending factorial (with n terms) e.g. $(m+r-1)_n = (m+r-1)(m+r-2)\ldots(m+r-n)$

$(10)_4 = 10.9.8.7$, and $(2)_3 = 2.1.0 = 0$

Substituting from equations (16c), (16d) and (15) in (13) we get,

$$\sum_{m=1}^{\infty} C_{m,r+1} (m+r-1)_4 \, x^{m+r-5}$$
$$- \alpha \sum_{m=1}^{\infty} C_{m,r+1} [(m+r-1)_4 + 2(m+r-1)_3] x^{m+r-4}$$
$$- \lambda \sum_{m=1}^{\infty} C_{m,r+1} x^{m+r-1} = 0 \tag{17}$$

Since equation (17) is satisfied identically for all values of x, the coefficients of all powers of x should be zero. Equating the coefficient of lowest power of x to zero, we get the indicial equation

$$C_{1,r+1} \, r(r-1)(r-2)(r-3) = 0$$

Therefore we have the roots of indicial equation as

$$r = 0, 1, 2 \text{ and } 3$$

Wayland (9) suggests that when the roots of the indicial equation differ by an integer, the solution with lowest root be tried first. If this solution yields more than one arbitrary parameter then the portion of the solution involving each parameter represents one linearly independent solution which corresponds to the solution from higher root (differing by an integer). If however, this solution for the lowest root does not turn out to be a finite solution then the method of variation of parameters (7) has to be resorted to, starting from the highest root.

Equating the coefficients of higher powers of x in equation

(17), we get the recurrence relations for the coefficients of the infinite series,

$$C_{m,r+1} = \frac{\alpha\, C_{m-1,r+1}\left[(m+r-2)_4 + 2(m+r-2)_3\right]}{(m+r-1)_4}$$

$$= \frac{\alpha\, C_{m-1,r+1}\ (m+r-2)_3\ (m+r-3)}{(m+r-1)_4}, \text{ for } m = 2,\ 3,\ 4 \qquad (18)$$

and
$$C_{m,r+1} = \frac{\lambda\, C_{m-4,r+1}}{(m+r-1)_4} + \frac{\alpha\, C_{m-1,r+1}\ (m+r-2)_3\ (m+r-3)}{(m+r-1)_4} \quad \text{for all } m \geqq 5 \qquad (19)$$

1. Consider the solution corresponding to $r=0$

we have $U_1 = \sum_{m=1}^{\infty} C_{m,1}\, x^{m-1}$

where $C_{m,1}$ for $m = 2, 3, \ldots, \infty$ are given by equations (18) or (19). And C_{11}, which is arbitrary, be 1.

$$C_{21} = \frac{\alpha(1)(0)_3(-1)}{(1)_4} = \text{Indeterminate}$$

$$C_{31} = \frac{\alpha C_{21}(1)_3(0)}{(2)_4} = \text{Indeterminate}$$

$$C_{41} = \frac{\alpha C_{31}(2)_3(1)}{(3)_4} = \text{Indeterminate}$$

$$C_{51} = \frac{\lambda\ C_{11}}{(4)_4} + \frac{\alpha\ C_{41}(3)_3(2)}{(4)_4}$$

$$C_{61} = \frac{\lambda\ C_{21}}{(5)_4} + \frac{\alpha\ C_{51}(4)_3(3)}{(5)_4} \quad \text{and so on.}$$

Thus, this solution involves C_{11}, C_{21}, C_{31}, and C_{41} as arbitrary constants. In fact, C_{21} in this solution (coefficient of x) corresponds to C_{12} (coefficient of x in the solution with r=1). C_{31} corresponds to C_{13} (coefficients of x^2 term) and C_{41} to C_{14}. For computing the coefficients on a computer. however, it is easier to equate C_{21}, C_{31}, C_{41} to zero and generate the solutions corresponding to these seperately for the roots r = 1,2 and 3 from the recurrence relations (18) and (19).

Thus the recurrence relations (18) and (19) give four solutions,

$$U_1 = \sum_{m=1}^{\infty} C_{m,1}\, x^{m-1} \qquad \text{corresponding to the root } r = 0$$

$$U_2 = \sum_{m=1}^{\infty} C_{m,2}\, x^{m} \qquad \text{corresponding to the root } r = 1$$

$$U_3 = \sum_{m=1}^{\infty} C_{m,3}\, x^{m+1} \qquad \text{corresponding to the root } r = 2$$

and

$$U_4 = \sum_{m=1}^{\infty} C_{m,4}\, x^{m+2} \qquad \text{corresponding to the root } r = 3$$

The C's are given by equations (18) and (19) and the indeterminate terms equated to zero for computational ease.

It turns out that the coefficients C_{12}, C_{13}, C_{14}, C_{22}, C_{23}, C_{24}, C_{32}, C_{33} and C_{34} are zeros.

Now the general solution to equation (13) is

$$U = \sum_{n=1}^{4} A_n U_n \tag{20}$$

where the A_n are determined by the boundary conditions.

Simply Supported Beam:

Substituting the boundary conditions (8) in equation (20) we get,

$$A_1U_1(0) + A_2\cancel{U_2}(0) + A_3\cancel{U_3}(0) + A_4\cancel{U_4}(0) = 0 \tag{21a}$$

$$A_1\cancel{U_1''}(0) + A_2\cancel{U_2''}(0) + A_3U_3''(0) + A_4\cancel{U_4''}(0) = 0 \tag{21b}$$

$$\cancel{A_1}U_1(1) + A_2U_2(1) + \cancel{A_3}U_3(1) + A_4U_4(1) = 0 \tag{21c}$$

$$\cancel{A_1}U_1''(1) + A_2U_2''(1) + \cancel{A_3}U_3''(1) + A_4U_4''(1) = 0 \tag{21d}$$

where $U_n(k)$ represents U_n at $x = k$, and $/$ indicates, that the term goes to zero.

Equations (21a) and (21b) give $A_1 = A_3 = 0$. Subsequently, equations (21c) and (21d) yield the characteristic equation,

$$\begin{vmatrix} U_2(1) & U_4(1) \\ U_2''(1) & U_4''(1) \end{vmatrix} = 0$$

where U_2 and U_4 are functions of α, λ, and x. And the solutions to equation (12) is

$$\begin{aligned} U &= A_2U_2 + A_4U_4 \\ &= A_4\left[U_4 - \frac{U_4(1)}{U_2(1)}\,U_2\right] \end{aligned} \tag{23}$$

For a particular value of α, the λ's satisfying equation (22) are the eigenvalues of the problem and equation (23) gives eigenvectors corresponding to the respective eigenvalues.

CHAPTER IV

THE SOLUTION AND RESULTS

The problem was solved on IBM 360/50 system. The computer program (refer appendix) consists of a function subprogram FUN that evaluates the value of characteristic equation and two subroutine subprograms, ROOT for finding the eigenvalues (roots of truncated infinite series) and EIGEN for giving the mode shape, given the eigenvalue. The subprogram FUN evaluates all the coefficients of the series solutions until they reduce to 10^{-20} in absolute value. Since the coefficients grow in magnitude with increasing values of λ, more and more terms of the power series will be necessary at higher frequencies for maintaining the desired accuracy of the solution. This subprogram cannot be of a general nature since the recurrence expression is decided by the differential equation formulation and the characteristic equation takes its shape from the boundary conditions of the problem. Nevertheless, for changes in boundary conditions, necessary changes in the program are fairly easy. The subprograms FUN and EIGEN included in the appendix are for linear temperature variation along a simply supported beam. The cards that need to be changed for different temperature variations and boundary conditions carry an identification in columns 73 through 80. The variable names and subscripts correspond very closely to those in this report and therefore the program is self explanetory.

The subroutine ROOT is quite general in nature. It gives the roots of any function, given an initial value, increment, final value and accuracy desired. It also prints out values of the function at various values of the argument during its root finding process.

The subroutine EIGEN evaluates the coefficients of the series in the same manner as does the FUN, but instead of evaluating the value of characteristic equation, it evaluates the values of U at various x values according to equation (23).

The first three eigenvalues, corresponding to the first three modes of vibration, were obtained for α values of 0.25, 0.50 and 0.75. Table I gives first three eigenvalues for various values of α. The eigenvalue λ being equal to $\frac{\Omega^2 L^4 \rho}{E_o I}$. Fig. 4a, which is a graph of $\lambda_1(\alpha)$ divided by $\lambda_1(0)$ Vs. α, shows the effect of the temperature gradient on the first natural frequency of the simply supported beam. Fig. 4b and 4c show the effect of α on the second and third frequency of the simply supported beam. As expected, the frequency lowers with decrease in E i.e. increase in α.

The eigenvectors, for various α values are tabulated in Tables II. The mode shapes are normalised with respect to a convenient element so that the effect of temperature variation can be readily noticed. The mode shapes for $\alpha = 0.50$ are plotted in Fig. 5. The mode shape gives a good indication of the order of frequency. This is essential since it is possible to miss two consecutive eigenvalues in the root finding process of the

ROOT subroutine, if the two eigenvalues lie within the prescribed interval.

Double precision arithmatic is used in working the problem. The difference in results from single precision to double precision is very significant at higher frequencies. It is interesting to see that, the percentage reduction due to the temperature variation is almost same for at least the first three eigenvalues. Therefore the graphs in Fig. 4a, 4b & 4c look similar. The results tabulated in Table I compare very well with those obtained by Marangoni, Fauconneau and Scipio (1) using bounding techniques. The solution by Frobenius' method requires about half the amount of work as the bounding techniques with the additional advantage of being a very straightforward method.

Table I

Effect of modulus variation on the first three eigenvalues of a simply supported beam

Order of Eigenvalue	1	2	3
$\alpha = 0.0$	97.409091	1558.5455	7890.1364
$\alpha = 0.25$	84.989636	1358.6533	6876.2367
$\alpha = 0.50$	71.901079	1144.7606	5785.9472
$\alpha = 0.75$	57.584701	904.76278	4551.9378



a)



Fig. 4a, 4b, 4c. Effect of modulus variation on first three eigenvalues of simply supported beam.

Table II a

First mode shape of a simply supported beam for different temperature gradients

Deflection at	$\alpha = 0.0$	$\alpha = 0.25$	$\alpha = 0.50$	$\alpha = 0.75$
x = 0.0	0.0	0.0	0.0	0.0
x = 0.1	-0.309	-0.300	-0.289	-0.278
x = 0.2	-0.588	-0.574	-0.560	-0.537
x = 0.3	-0.809	-0.798	-0.778	-0.758
x = 0.4	-0.951	-0.946	-0.932	-0.919
x = 0.5	-1.000	-1.000	-1.000	-1.000
x = 0.6	-0.951	-0.960	-0.970	-0.988
x = 0.7	-0.809	-0.825	-0.842	-0.875
x = 0.8	-0.588	-0.601	-0.620	-0.660
x = 0.9	-0.309	-0.318	-0.331	-0.358
x = 1.0	-0.000	-0.000	-0.000	-0.000



Fig. 5a. First mode shape of a simply supported beam for $\alpha = 0.50$.

Table II b

Second mode shape of a simply supported beam for different temperature gradients

Deflection at	$\alpha = 0.0$	$\alpha = 0.25$	$\alpha = 0.50$	$\alpha = 0.75$
x = 0.0	0.0	0.0	0.0	0.0
x = 0.1	-0.618	-0.595	-0.568	-0.535
x = 0.2	-1.000	-0.977	-0.948	-0.911
x = 0.3	-1.000	-1.000	-1.000	-1.000
x = 0.4	-0.618	-0.649	-0.688	-0.745
x = 0.5	0.0	-0.046	-0.110	-0.205
x = 0.6	0.618	0.584	0.532	0.449
x = 0.7	1.000	0.996	0.987	0.966
x = 0.8	1.000	1.023	1.055	1.110
x = 0.9	0.618	0.641	0.679	0.751
x = 1.0	0.000	0.000	0.000	0.000



Fig. 5b. Second mode shape of a simply supported beam for $\alpha = 0.50$.

Table II c

Third mode shape of a simply supported beam for different temperature gradients

Deflection at	$\alpha = 0.0$	$\alpha = 0.25$	$\alpha = 0.50$	$\alpha = 0.75$
x = 0.0	0.0	0.0	0.0	0.0
x = 0.1	-0.809	-0.785	-0.764	-0.753
x = 0.2	-0.951	-0.956	-0.970	-1.011
x = 0.3	-0.309	-0.365	-0.441	-0.561
x = 0.4	0.588	0.528	0.447	0.323
x = 0.5	1.000	1.000	1.000	1.000
x = 0.6	0.588	0.651	0.741	0.894
x = 0.7	-0.309	-0.250	-0.161	0.001
x = 0.8	-0.951	-0.949	-0.953	-0.959
x = 0.9	-0.809	-0.841	-0.899	-1.033
x = 1.0	-0.000	-0.000	-0.000	-0.000



Fig. 5c. Third mode shape of a simply supported beam for $\alpha = 0.50$.

CHAPTER V

ILLUSTRATIVE CASES

1. Beam with Clamped Ends (linear temperature variation)

Imposing the boundary conditions (9) from page 5 on equation (20) appearing on page 12, we get,

$$\begin{aligned}
A_1U_1(0) + A_2U_2(0) + A_3U_3(0) + A_4U_4(0) &= 0\\
A_1U_1'(0) + A_2U_2'(0) + A_3U_3'(0) + A_4U_4'(0) &= 0\\
A_1U_1(1) + A_2U_2(1) + A_3U_3(1) + A_4U_4(1) &= 0\\
A_1U_1'(1) + A_2U_2'(1) + A_3U_3'(1) + A_4U_4'(1) &= 0
\end{aligned} \tag{24}$$

from equations (24) we get the characteristic equation,

$$\begin{vmatrix} U_3(1) & U_4(1) \\ U_3'(1) & U_4'(1) \end{vmatrix} = 0 \tag{25}$$

and the mode shape can be expressed as

$$\begin{aligned}
U &= A_3U_3 + A_4U_4\\
&= A_4\left[U_4 - \frac{U_4(1)}{U_3(1)} U_3 \right]
\end{aligned} \tag{26}$$

First three eigenvalues for $\alpha = 0.25$, 0.50 and 0.75 were calculated and are tabulated in Table III. The first three mode shapes for various α values are tabulated in Table IV. Fig. 7 shows the mode shapes for $\alpha = 0.50$.

Table III

Effect of modulus variation on the first three eigenvalues of a beam with clamped ends

Order of Eigenvalue	1	2	3
$\alpha = 0.0$	500.56393	3803.5370	14617.630
$\alpha = 0.25$	435.76258	3312.5286	12732.884
$\alpha = 0.50$	364.75663	2778.2510	10688.285
$\alpha = 0.75$	281.72376	2160.3328	8335.5640



a)



b)



c)

Fig. 6a, 6b, 6c. Effect of modulus variation on first three eigenvalues of a beam with clamped ends.

Table IV a

First mode shape of a beam with clamped ends for different temperature gradients

Deflection at	$\alpha = 0.0$	$\alpha = 0.25$	$\alpha = 0.50$	$\alpha = 0.75$
x = 0.0	0.0	0.0	0.0	0.0
x = 0.1	-0.119	-0.111	-0.101	-0.090
x = 0.2	-0.390	-0.370	-0.344	-0.313
x = 0.3	-0.691	-0.666	-0.634	-0.592
x = 0.4	-0.916	-0.900	-0.877	-0.845
x = 0.5	-1.000	-1.000	-1.000	-1.000
x = 0.6	-0.916	-0.934	-0.959	-1.001
x = 0.7	-0.691	-0.717	-0.758	-0.830
x = 0.8	-0.390	-0.414	-0.450	-0.522
x = 0.9	-0.119	-0.129	-0.145	-0.179
x = 1.0	-0.000	-0.000	-0.000	-0.000



Fig. 7a. First mode shape of a beam with clamped ends for $\alpha = 0.50$

Table IV b

Second mode shape of a beam with clamped ends
for different temperature gradients

Deflection at	$\alpha = 0.0$	$a = 0.25$	$\alpha = 0.50$	$\alpha = 0.75$
x = 0.0	0.0	0.0	0.0	0.0
x = 0.1	-0.303	-0.285	-0.265	-0.240
x = 0.2	-0.802	-0.774	-0.741	-0.699
x = 0.3	-1.000	-1.000	-1.000	-1.000
x = 0.4	-0.688	-0.734	-0.793	-0.881
x = 0.5	0.0	-0.068	-0.163	-0.316
x = 0.6	0.688	0.644	0.577	0.453
x = 0.7	1.000	1.006	1.012	1.009
x = 0.8	0.802	0.839	0.897	1.001
x = 0.9	0.303	0.326	0.366	0.451
x = 1.0	0.000	0.000	0.000	0.000



Fig. 7b. Second mode shape of a beam with clamped ends for $\alpha = 0.50$.

Table IV c

Third mode shape of a beam with clamped ends
for different temperature gradients

Deflection at	$\alpha = 0.0$	$\alpha = 0.25$	$\alpha = 0.50$	$\alpha = 0.75$
x = 0.0	0.0	0.0	0.0	0.0
x = 0.1	-0.548	-0.519	-0.493	-0.476
x = 0.2	-1.073	-1.060	-1.058	-1.097
x = 0.3	-0.618	-0.683	-0.773	-0.939
x = 0.4	0.447	0.361	0.245	0.053
x = 0.5	1.000	1.000	1.000	1.000
x = 0.6	0.447	0.536	0.666	0.899
x = 0.7	-0.617	-0.552	-0.451	-0.257
x = 0.8	-1.072	-1.093	-1.136	-1.227
x = 0.9	-0.549	-0.585	-0.659	-0.838
x = 1.0	-0.000	-0.000	-0.000	-0.000



Fig. 7c. Third mode shape of a beam with clamped ends for $\alpha = 0.50$.

2. Beam with Free Ends (Linear temperature variation)

The boundary conditions (9) for a beam with free ends are,

$$U''(0) = U''(1) = 0 \tag{27a}$$

and $\frac{d}{dx}\left[(1-\alpha x)\frac{d^2U}{dx^2}\right] = 0$ at $x = 0$ and $x = 1$

i.e. $-\alpha U'' + (1-\alpha x)U''' = 0$ at $x = 0$ and $x = 1$

Since $U''(0)$ and $U''(1)$ are zeros and $(1-\alpha x) \neq 0$, we get,

$$U'''(0) = U'''(1) = 0 \tag{27b}$$

Substituting the boundary conditions (27) in (20) we get,

$$\begin{aligned}
A_1U_1''(0) + A_2U_2''(0) + A_3U_3''(0) + A_4U_4''(0) &= 0\\
A_1U_1'''(0) + A_2U_2'''(0) + A_3U_3'''(0) + A_4U_4'''(0) &= 0\\
A_1U_1''(1) + A_2U_2''(1) + A_3U_3''(1) + A_4U_4''(1) &= 0\\
A_1U_1'''(1) + A_2U_2'''(1) + A_3U_3'''(1) + A_4U_4'''(1) &= 0
\end{aligned} \tag{28}$$

from equations (28) we get the characteristic equation,

$$\begin{vmatrix} U_1''(1) & U_2''(1) \\ U_1'''(1) & U_2'''(1) \end{vmatrix} = 0 \tag{29}$$

The mode shape is given by

$$\begin{aligned}
U &= A_1U_1 + A_2U_2\\
&= A_2\left[U_2 - \frac{U_2''(1)}{U_1''(1)}U_1\right]
\end{aligned} \tag{30}$$

Table V gives first three eigenvalues for $\alpha = 0.25$, 0.50, and 0.72. Fig. 9 gives the mode shapes for $\alpha = 0.50$.

Table V

Effect of modulus variation on the first three eigenvalues of a beam with free ends

Order of Eigenvalue	1	2	3
$\alpha = 0.0$	500.56393	3803.5370	14617.630
$\alpha = 0.25$	437.08963	3317.5037	12743.582
$\alpha = 0.50$	371.14705	2802.3870	10740.483
$\alpha = 0.75$	300.82009	2234.2973	8498.7080



a)



Fig. 8a, 8b, 8c. Effect of modulus variation on first three eigenvalues of a beam with free ends.

Table VI a

First mode shape of a beam with free ends for different temperature gradients

Deflection at	$\alpha = 0.0$	$\alpha = 0.25$	$\alpha = 0.50$	$\alpha = 0.75$
x = 0.0	-1.646	-1.617	-1.585	-1.546
x = 0.1	-0.884	-0.878	-0.875	-0.864
x = 0.2	-0.161	-0.172	-0.186	-0.206
x = 0.3	0.448	0.429	0.405	0.373
x = 0.4	0.856	0.842	0.824	0.800
x = 0.5	1.000	1.000	1.000	1.000
x = 0.6	0.856	0.870	0.890	0.921
x = 0.7	0.448	0.467	0.496	0.543
x = 0.8	-0.161	-0.148	-0.130	-0.098
x = 0.9	-0.884	-0.891	-0.901	-0.920
x = 1.0	-1.646	-1.676	-1.726	-1.815



Fig. 9a. First mode shape of a beam with free ends for $\alpha = 0.50$.

Table VI b

Second mode shape of a beam with free ends for different temperature gradients

Deflection at	$\alpha = 0.0$	$\alpha = 0.25$	$\alpha = 0.50$	$\alpha = 0.75$
x = 0.0	-1.510	-1.505	-1.507	-1.523
x = 0.1	-0.344	-0.366	-0.396	-0.441
x = 0.2	0.600	0.572	0.537	0.488
x = 0.3	1.000	1.000	1.000	1.000
x = 0.4	0.729	0.770	0.824	0.902
x = 0.5	0.0	0.053	0.130	0.249
x = 0.6	-0.729	-0.706	-0.671	-0.602
x = 0.7	-1.000	-1.025	-1.062	-1.113
x = 0,8	-0.600	-0.646	-0.718	-0.842
x = 0.9	0.344	0.330	0.308	0.261
x = 1.0	1.510	1.560	1.643	1.800



Fig. 9b. Second mode shape of a beam with free ends for $\alpha = 0.50$.

Table VI c

Third mode shape of a beam with free ends for different temperature gradients

Deflection at	α = 0.0	α = 0.25	α = 0.50	α = 0.75
x = 0.0	-1.406	-1.388	-1.385	-1.427
x = 0.1	0.073	0.042	0.001	-0.056
x = 0.2	0.904	0.884	0.865	0.859
x = 0.3	0.558	0.607	0.675	0.785
x = 0.4	-0.461	-0.390	-0.293	-0.144
x = 0.5	-1.000	-1.000	-1.000	-1.000
x = 0.6	-0.461	-0.536	-0.644	-0.826
x = 0.7	0.558	0.507	0.432	0.291
x = 0.8	0.905	0.930	0.977	1.068
x = 0.9	0.077	0.109	0.167	0.287
x = 1.0	-1.401	-1.439	-1.510	-1.691



Fig. 9c. Third mode shape of a beam with free ends α = 0.50.

CHAPTER VI

CONCLUSION

Graphs such as Fig. 4, 6 & 8 are of immediate importance to designers. Knowing the frequency of vibration of a uniform beam, one can obtain directly the frequency corresponding to the particular α value for a problem. When temperature distributions other than the few analysed in papers (1) are encountered, Frobenius' method leads to a fairly straightforward and direct approach. One must use care when considering the radius of convergence of the solution. There are standard techniques to test the radius of convergence of the power series expansions of the functions a(x), the coefficient of next to the highest derivative; b(x), the coefficient of the derivative next to that, etc. (Ref. Equation (14) on page 8). The radius of convergence of the solution is the minimum of the radius of convergence of the power series expansions for a(x), b(x) etc.

When the variation of elastic modulus with temperature is not linear, we have to use the following expression to get E(x)

$$E(x) = \int \frac{\partial E(T)}{\partial T} \frac{\partial T(x)}{\partial x} \, dx + \text{Constant}$$

Which will give E(x) in the form $E_o(x) \cdot f(x)$ which can be substituted in the formulation (7)

LIST OF REFERENCES

1. Marangoni, Fauconneau & Scipio., "Effect of Non-homogeneity on the transverse vibrational frequency of uniform beams", Int. J. Engng. Sci., Vol. 6 pp. 637-659 (1968)

2. Walker H.S. and Huang C.L., "Vibration and Stability of Rockets", American Astronautical Society, Advances in Astronautical Sciences, 24: 5-8.(1967)

3. Meirovitch Leonard., Analytical Methods in Vibrations, The Macmillan Co. New York (1967)

4. Garrick I.E., "A survey of Thermoelasticity", Aerospace Engineering, Jan. 1963.

5. Spinner S., "Temperature Dependance of Elastic Constants of Some Cermet Specimens", J. of Research; U.S. Nat. Bur. of Standards, Vol. 65C, No. 2, Apr-June 1961

6. Hoff N.J., High Temperature effects in aircraft Structures, p. 61, Pergamon Press (1958)

7. Kreyszig E., Advanced Engineering Mathematics, pp. 155, John Wiley & Sons. Inc. (1966)

8. Coddington E.A., Ordinary Differential Equations, Prentice Hall Inc. (1961)

9. Wayland Harold., Differential Equations applied in Science and Engineering, p. 138, D. Van Nostrand Co. Inc. (1957)

APPENDIX

THE MAIN PROGRAM

```
C     VIBRATION ANALYSIS OF NON-HOMOGENEOUS UNIFORM BEAM
      IMPLICIT REAL * 8 (A-H,O-Z)
      REAL *8 LAMBDA
  101 FORMAT (1H1)
  110 FORMAT('1EIGENVALUE CORROSPONDING TO MODE ',I1,' AT',
     1 E16.8,',',15X,'ALPHA = ',F5.2///)
      ALPHA = 0.0
  200 ALPHA = ALPHA + 0.25
      LAMBDA = 1.0E-02
      KOUNT = 0
      DELI = 500.00
      DDEL = 0.25
      DELX = 0.01
      XMAX = 1.0E+06
  201 CALL ROOT (LAMBDA, DELI, DDEL, DELX, XMAX, 1, XS, XL,
     1YS, YL, G, ALPHA)
      KOUNT = KOUNT + 1
  215 WRITE (3,110) KOUNT, LAMBDA, ALPHA
      CALL EIGEN (LAMBDA, ALPHA)
      WRITE (3,101)
      LAMBDA = LAMBDA + 10.0
      IF (KOUNT.LT.3) GO TO 201
      IF ( ALPHA.LT.0.74 ) GO TO 200
      STOP
      END
```

```
      FUNCTION FUN (LAMBDA, ALPHA)
C
C     THIS SUBPROGRAM EVALUATES THE COEFFICIENTS OF THE POWER
C     SERIES SOLUTIONS TO THE DIFFERENTIAL EQUATION AND
C     EVALUATES THE RESIDUAL OF THE CHARACTERISTIC EQUATION
C     AT TRIAL EIGENVALUES
C
      IMPLICIT REAL * 8 (A-H,O-Z)
      REAL *8 LAMBDA
      COMMON C(1000,4)
  100 FORMAT (1H-)
  101 FORMAT (1H1)
  105 FORMAT ( 8E16.8)
      DO 300 K = 1,4
  300 C(1,K) = 1.0
      DO 305 M = 2,4
      DO 305 K = 1,3
  305 C(M,K) = 0.0
      K = 4
      DO 195 M = 2,4
  195 C(M,K) =  ALPHA * C(M-1,K) * (M+K-4) / (M+K-2)
C     ONLY THE SOLUTIONS U2 AND U4 ARE REQUIRED FOR THE SIMPLY
C     SUPPORTED BEAM
C
C     K IS EQUAL TO R + 1 IN THE RECURRENCE EXPRESSION
C     STATEMENTS 201 AND 203
      K = 2
      DO 200 M = 5,1000
  201 C(M,K) = LAMBDA*C(M-4,K)/((M+K-2)*(M+K-3)*(M+K-4)*
     1(M+K-5))+ ALPHA * C(M-1,K) * (M+K-4) / (M+K-2)
C     TEST IF COEFFICIENT HAS BECOME SUFFICIENTLY SMALL
      IF (DABS ( C(M,K) ).LT. 1.0E-20 ) GO TO 202
  200 CONTINUE
  202 NCOEFF = M
      K = 4
      DO 203 M = 5,NCOEFF
  203 C(M,K) = LAMBDA*C(M-4,K)/((M+K-2)*(M+K-3)*(M+K-4)*
     1(M+K-5))+ ALPHA * C(M-1,K) * (M+K-4) / (M+K-2)
  215 U2 = 0.0
      DO 350 M = 1,NCOEFF
  350 U2 = U2 + C(M,2)
      U4 = 0.0
      DO 360 M = 1,NCOEFF
  360 U4 = U4 + C(M,4)
      U2 2DOT = 0.0
      DO 370 M = 2,NCOEFF
  370 U2 2DOT = U2 2DOT + C(M,2) * M * (M-1)
      U4 2DOT = 0.0
      DO 380 M = 1,NCOEFF
  380 U4 2DOT = U4 2DOT + C(M,4) * (M+2) * (M+1)
      FUN = U2 * U4 2DOT - U4 * U2 2DOT
      RETURN
      END
```

```
      SUBROUTINE EIGEN (LAMBDA, ALPHA)
C
C     THIS SUBROUTINE EVALUATES THE MODE SHAPE,
C     GIVEN THE EIGENVALUE
C
      IMPLICIT REAL * 8 (A-H,O-Z)
      REAL *8 LAMBDA
      COMMON C(1000,4)
  100 FORMAT (1H-)
  101 FORMAT (1H1)
  102 FORMAT (8E16.8)
  105 FORMAT (//' AND THE CORRESPONDING EIGENVECTOR IS')
  106 FORMAT (// 10X,'U =',E16.8,5X,'AT X = ',F5.2 )
      DO 300 K = 1,4
  300 C(1,K) = 1.0
      DO 305 M = 2,4
      DO 305 K = 1,3
  305 C(M,K) = 0.0
      K = 4
      DO 195 M = 2,4
  195 C(M,K) =  ALPHA * C(M-1,K) * (M+K-4) / (M+K-2)
C
C     ONLY THE SOLUTIONS U2 AND U4 ARE REQUIRED FOR THE SIMPLY
C     SUPPORTED BEAM
C
C     K IS EQUAL TO R + 1 IN THE RECURRENCE EXPRESSION
C     STATEMENTS 201 AND 203
C
      K = 2
      DO 200 M = 5,1000
  201 C(M,K) = LAMBDA*C(M-4,K)/((M+K-2)*(M+K-3)*(M+K-4)*
     1(M+K-5))+ ALPHA * C(M-1,K) * (M+K-4) / (M+K-2)
C     TEST IF COEFFICIENT HAS BECOME SUFFICIENTLY SMALL
      IF (DABS ( C(M,K) ).LT. 1.0E-20 ) GO TO 202
  200 CONTINUE
  202 NCOEFF = M
      WRITE (3,102) (C(M,K),M = 1,NCOEFF)
      WRITE (3,100)
      K = 4
      DO 203 M = 5,NCOEFF
  203 C(M,K) = LAMBDA*C(M-4,K)/((M+K-2)*(M+K-3)*(M+K-4)*
     1(M+K-5))+ ALPHA * C(M-1,K) * (M+K-4) / (M+K-2)
      WRITE (3,102) (C(M,K),M = 1,NCOEFF)
  215 U2 = 0.0
      DO 350 M = 1,NCOEFF
  350 U2 = U2 + C(M,2)
      U4 = 0.0
      DO 360 M = 1,NCOEFF
  360 U4 = U4 + C(M,4)
      A2 = -U4/U2
  361 WRITE (3,105)
      X = -0.1
  400 X = X + 0.1
```

```
      U2 OF X = 0.0
      DO 410 M = 1,NCOEFF
      ADD = C(M,2) * X ** M
      IF ((DABS(ADD).LT.1.0E-20) .AND. (ADD.NE.0.0) ) GO TO 800
  410 U2 OF X = U2 OF X + ADD
  800 CONTINUE
      U4 OF X = 0.0
      DO 420 M = 1,NCOEFF
      ADD = C(M,4) * X **(M+2)
      IF ((DABS(ADD).LT.1.0E-20) .AND. (ADD.NE.0.0) ) GO TO 810
  420 U4 OF X = U4 OF X + ADD
  810 U OF X = U4 OF X + A2 * U2 OF X
      WRITE (3,106) U OF X, X
      IF ( X.LT..99 ) GO TO 400
      RETURN
      END
```

```
      SUBROUTINE ROOT(X,DELI,DDEL,DELX,XMAX,ILAST,XS,XL,YS,
     2YL,IPRINT,ALPHA)
C     THIS SUBROUTINE CALLS FUN (X,ALPHA) AND FINDS THE
C     ROOTS OF THE TRANSCENDENTAL EQUATION FUN(X,ALPHA)
C     IPRINT=1 PRINTS XS,XL,YS,YL.  IPRINT=0 DOES NOT PRINT.
C     DELI IS THE INITIAL INCREMENT
C     DDEL IS THE FACTOR FOR DIVIDING THE INTERVAL
C     DELX IS THE ACCURACY ON THE SOLUTION
C     XMAX IS THE MAXIMUM THAT THE ARGUMENT X MAY BECOME
C     ILAST IS THE NUMBER OF ROOTS DESIRED
      IMPLICIT REAL * 8 (A-H,O-Z)
    4 FORMAT (////4E20.8///)
  105 FORMAT ('-RESIDUAL OF THE CHARACTERISTIC EQUATION =',
     1E16.8,15X,'AT LAMBDA =',E16.8)
      IUI=0
      X=X-DELI
   10 DEL = DELI
      X = X + DEL
      Y = FUN (X, ALPHA )
      WRITE (3,105) Y,X
      IF(Y) 98,500,99
   98 ZAZ=-1.0
      GO TO 100
   99 ZAZ=+1.0
  100 XS = X
      YS = Y
      IF(X.GT.XMAX)  GO TO 510
      X = X + DEL
      Y = FUN (X, ALPHA )
      WRITE (3,105) Y,X
      IF(ZAZ*Y) 110,500,100
  110 XL = X
      YL = Y
  120 X = XS - YS*((XL-XS)/(YL-YS))
      IF(XL-XS-DELX) 500,130,130
  130 Y = FUN (X, ALPHA )
      WRITE (3,105) Y,X
      DEL = DEL*DDEL
      IF(ZAZ*Y) 140,500,100
  140 XL = X
      YL = Y
      X = X - DEL
      Y = FUN (X, ALPHA )
      WRITE (3,105) Y,X
      IF(ZAZ*Y) 140,500,150
  150 XS = X
      YS = Y
      GO TO 120
  500 IUI=IUI+1
      IF(IPRINT.EQ.0) GO TO 900
      WRITE(3,4)  XS,XL,YS,YL
  900 CONTINUE
  502 IF(ILAST-IUI) 510,510,10
  510 CONTINUE
      RETURN
```

ACKNOWLEDGEMENTS

I wish to express my sincere thanks to Dr. Hugh S. Walker, Associate Professor of Mechanical Engineering, for his guidance and encouragement throughout the preparation of this report. My thanks are also due to Dr. C. L. Huang, Associate Professor of Applied Mechanics, for his valuable help. I also wish to acknowledge for the computing facilities provided by the Kansas State University.

VIBRATION ANALYSIS OF NON-HOMOGENEOUS BEAMS

by

PARSHURAM GANESH DATE

B.E(Mech)., Marathwada University
Aurangabad, India 1964

AN ABSTRACT OF A MASTER'S REPORT

submitted in partial fulfillment of the

requirements for the degree

MASTER OF SCIENCE

Department of Mechanical Engineering

KANSAS STATE UNIVERSITY
Manhattan, Kansas

1969

Name : Parshuram Ganesh Date

Date of Degree : August, 1969

Institution : Kansas State University

Location : Manhattan, Kansas

Title of Study : VIBRATION ANALYSIS OF NON-HOMOGENEOUS BEAMS

Pages in Study : 40

Candidate for Degree of Master of Science

Major Field : Mechanical Engineering

Scope and Method of Study: Non-homogeneity caused by a linear temperature variation along the length of a beam is considered. Frobenius' method is used to obtain a power series solution to the linear, non-constant coefficient differential equation formulation. The coefficients of the power series are obtained on a computer. Various and conditions for beams with various temperature gradients are analysed.

Findings and Conclusion: As expected, the vibrational frequencies decrease with increasing temperature gradients. The Frobenius' method leads to a straightforward solution of a linear, non-constant coefficient differential equation formulation. Problems involving variation in cross section, density etc. lead to the same type of mathematical model as this. This method is particularly useful with computers since increased accuracy is obtained by using more terms in the power series solution.

MAJOR PROFESSOR'S APPROVAL Hugh S. Walker

VITA

Parshuram Ganesh Date

Candidate for the Degree of

Master of Science

Report: VIBRATION ANALYSIS OF NON-HOMOGENEOUS BEAMS

Major Field: Mechanical Engineering

Biographical:

Personal Data: Born in Hyderabad, Andhra Pradesh, India, February 26, 1943, the son of Shri. and Smt. Ganesh Vaman Date.

Education: Graduated from Vivek Vardhini High School, Hyderabad, India; received the Bachelor of Engineering degree from Marathwada University, Aurangabad, India, with a major in Mechanical Engineering, in June 1964; completed requirements for the Master of Science degree from Kansas State University, with a major in Mechanical Engineering, in August 1969.

Professional Experience: Worked as a Scientific Officer at the Atomic Energy Establishment Trombay, Govt. of India from Aug. 1964 to Aug. 1968. Of which, the last $1\frac{1}{2}$ years in the capacity of Shift Engineer at the Canada-India Reactor, a 40,000 Kw. research reactor plant.